```
In[ ]:= Style[DateString[], 15, ColorData[59, 1], Bold, FontFamily → "Helvetica"]
Out[ ]= Sun 13 Mar 2022 20:24:17

In[61]:= Import["Header.jpg"]
```

Mathematica notebook

Dr. Martin Ricker
Instituto de Biología, Universidad Nacional Autónoma de México (UNAM), Mexico City
mricker@ib.unam.mx, martin_tuxtlas@yahoo.com.mx

The first function (“confidenceIntervalsForProportionsFunction”) calculates simultaneous 95% confidence intervals for several proportions. Each proportion is calculated for one cohort with an initial and a final number of individuals:

The input consists of two numbers for each of one to several cohorts: the initial and the final number of individuals. The initial number of individuals may vary among cohorts. The method used for calculating confidence intervals for proportions is explained in Fleiss et al. (2003: 25). The Bonferroni method is used to adjust confidence intervals for simultaneous inspection. It is explained, for example, in Sokal & Rohlf (2012: 239).

Fleiss, J.L., Levin, B., & Paik, M.C. (2003). *Statistical Methods for Rates and Proportions* (3rd edition). Hoboken, New Jersey: John Wiley & Sons, 760 pages.

Sokal, R.R., & Rohlf, F.J. (2012). Biometry (4th edition). New York, USA: W.H. Freeman and Company. 937 pages.

Supplementary matherial of the original: Mineral fertilization in tropical reforestation: seedling growth and survival of ten native tree species in Tabasco, Mexico.
Received: 24 September 2020 Accepted: 15 November 2021 Published: 17 June 2022

This document must be cited as:
Martínez-Bravo, R. D., Ricker, M., Nava-Cruz, Y., & Siebe, C. (2022). Mineral fertilization in tropical reforestation: seedling growth and survival of ten native tree species in Tabasco, Mexico. *Madera y Bosques*, *28*(1), e2812269. https://doi.org/10.21829/myb.2022.2812269. Supplement 2.

```
In[ ]:= confidenceIntervalsForProportionsFunction =
  Function[groupedZeroOneData,
   nCohorts = Length[groupedZeroOneData];
   If[nCohorts == 1, Print["95% confidence interval for 1 cohort:"]];
   If[nCohorts > 1,
    Print["Simultaneous 95% confidence intervals for ", nCohorts, " cohorts:"]];
   Print["Lower limit, Proportion, Upper limit, Initial number"];
   Table[
    x0 = groupedZeroOneData[[i, 1]];
    nn = groupedZeroOneData[[i, 2]];
    alphaAdj = 0.05 / nCohorts;
    If[x0 == 0, PL = 0,
     FLvalue = InverseCDF[FRatioDistribution[2 * (nn - x0 + 1), 2 * x0], 1 - alphaAdj];
     PL = x0 / (x0 + (nn - x0 + 1) * FLvalue)];
    If[x0 == nn, PU = 1,
     FUvalue = InverseCDF[FRatioDistribution[2 * (x0 + 1), 2 * (nn - x0)], 1 - alphaAdj];
     PU = (x0 + 1) * FUvalue / ((nn - x0) + (x0 + 1) * FUvalue)];
    {PL, N[x0 / nn], PU, nn}, {i, 1, nCohorts}]];
```

Following, in the first example, there are 4 cohorts. In the second example, there is only 1 cohort.

```
In[ ]:= confidenceIntervalsForProportionsFunction[{{7, 60}, {11, 60}, {4, 61}, {6, 61}}]

Simultaneous 95% confidence intervals for 4 cohorts:

Lower limit, Proportion, Upper limit, Initial number

Out[ ]= {{0.0418797, 0.116667, 0.242077, 60}, {0.0859582, 0.183333, 0.321928, 60},
  {0.0146863, 0.0655738, 0.17414, 61}, {0.0315804, 0.0983607, 0.217626, 61}}

In[ ]:=
confidenceIntervalsForProportionsFunction[{{7, 60}}]

95% confidence interval for 1 cohort:

Lower limit, Proportion, Upper limit, Initial number

Out[ ]= {{0.0560549, 0.116667, 0.207991, 60}}

In[ ]:=
```

The second function ("confidenceIntervalsForMediansFunction") calculates simultaneous 95% confidence intervals for medians of several statistical groups:

The input consists of data for one to several groups, for each of which the median is calculated. The method to calculate confidence intervals for medians is described in Higgins (2004: 13-14). The groups can have different numbers of data points. The Bonferroni method is used to adjust confidence intervals for simultaneous inspection. It is explained, for example, in Sokal & Rohlf (2012: 239). It is explained, for example, in Sokal & Rohlf (2012: 239).

Higgins, J.J. (2004). Introduction to Modern Nonparametric Statistics. Pacific Grove, California, USA: Brooks/Cole - Thomson Learning. 366 pages.

Sokal, R.R., & Rohlf, F.J. (2012). Biometry (4th edition). New York, USA: W.H. Freeman and Company. 937 pages.

```
In[ ]:= confidenceIntervalsForMediansFunction =
  Function[groupedData, nGroups = Length[groupedData];
   alphaAdj = 0.05 / nGroups;
   zz = InverseCDF[NormalDistribution[], 1 - alphaAdj];
   positionsWithNormalDistr = Table[nn = Length[groupedData[[i]]];
     lowerPos0 = Round[(nn - Sqrt[nn] * zz) / 2];
     upperPos0 = Round[(2 + nn + Sqrt[nn] * zz) / 2];
     If[lowerPos0 < 1, lowerPos = 1, lowerPos = lowerPos0];
     If[upperPos0 > nn, upperPos = nn, upperPos = upperPos0];
     {lowerPos, upperPos, nn}, {i, 1, nGroups}];
   solution = Table[possiblePositions =
       {positionsWithNormalDistr[[i, 3]], positionsWithNormalDistr[[i, 1 ;; 2]],
        If[positionsWithNormalDistr[[i, 1]] - 1 ≥ 1, {positionsWithNormalDistr[[i, 1]] - 1,
          positionsWithNormalDistr[[i, 2]]}, positionsWithNormalDistr[[i]]],
        {positionsWithNormalDistr[[i, 1]] + 1, positionsWithNormalDistr[[i, 2]]},
        {positionsWithNormalDistr[[i, 1]], positionsWithNormalDistr[[i, 2]] - 1},
        If[positionsWithNormalDistr[[i, 2]] + 1 ≤ positionsWithNormalDistr[[i, 3]],
         {positionsWithNormalDistr[[i, 1]], positionsWithNormalDistr[[i, 2]] + 1},
         positionsWithNormalDistr[[i]]], {positionsWithNormalDistr[[i, 1]] + 1,
         positionsWithNormalDistr[[i, 2]] - 1}, If[(positionsWithNormalDistr[[i, 1]] - 1 ≥ 1) &&
          (positionsWithNormalDistr[[i, 2]] + 1 ≤ positionsWithNormalDistr[[i, 3]]),
         {positionsWithNormalDistr[[i, 1]] - 1, positionsWithNormalDistr[[i, 2]] + 1},
         positionsWithNormalDistr[[i]]]};
     nn = possiblePositions[[1]];
     alphasCalc = Table[{k1 = possiblePositions[[j, 1]], k2 = possiblePositions[[j, 2]],
        N[1 - Total[Table[nn ! / (l ! * (nn - l) !), {l, k1, k2}] / 2^nn]]},
       {j, 2, Length[possiblePositions]}];
     Sort[Transpose[{Transpose[alphasCalc][[1]],
         Transpose[alphasCalc][[2]], (alphaAdj - Transpose[alphasCalc][[3]])}],
       Abs[#1[[-1]]] < Abs[#2[[-1]]] &][[1]], {i, 1, nGroups}];
   If[nGroups == 1, Print["95% confidence interval for 1 group:"]];
   If[nGroups > 1,
    Print["Simultaneous 95% confidence intervals for ", nGroups, " groups:"]];
   Print["Lower limit, Median, Upper limit, Initial number, ",
    Subscript[α, "adj."] - Subscript[α, "calc."]];
   Table[{groupedData[[i, solution[[i, 1]]]], N[Median[groupedData[[i]]]], groupedData[[i,
       solution[[i, 2]]]], Length[groupedData[[i]]], solution[[i, 3]]}, {i, 1, nGroups}]];
```

In the first example, there are 4 different-sized groups. The output consists of the lower and upper confidence limits, the number of data points in the group, and the difference between the adjusted and calculated alpha (where alpha is the significance level). The upper and lower confidence limits correspond to actual data values in the sample. According to equation (2) of the manuscript, one has to find the two positions “k1” and “k2” in the ordered sample that lead to the desired alpha, and the corresponding data values are the resulting limits. Since the positions in the ordered sample are integer numbers, and the number of data in the group is limited, it is generally not possible to get the desired exact alpha. In the second example, there is only 1 group.

In[ ]:=

```
data = {{0.5, 1, 2, 2.5, 3, 3, 3, 3.5, 5.5, 6, 7,
    7, 7.5, 7.5, 8.5, 9.5, 10, 10, 10.5, 12, 12, 18, 18, 19, 24, 25, 30},
   {2.5, 5, 6.5, 7.5, 7.5, 8.5, 10.5, 11, 12, 13.5, 14, 17, 21, 22, 22, 23.5, 23.5},
   {1.5, 3.5, 4.5, 7.5, 8, 9.5, 12.5, 16, 16, 18, 43.5}, {5.5, 9, 18.5}};

confidenceIntervalsForMediansFunction[data]
```

Simultaneous 95% confidence intervals for 4 groups:

Lower limit, Median, Upper limit, Initial number, $\alpha_{adj.} - \alpha_{calc.}$

Out[ ]= {{3.5, 7.5, 12, 27, -0.0000409514}, {7.5, 12., 21, 17, -0.00022583},
 {3.5, 9.5, 16, 11, 0.00078125}, {5.5, 9., 18.5, 3, -0.1125}}

In[ ]:=

```
confidenceIntervalsForMediansFunction[{{1.5, 3.5, 4.5, 7.5, 8, 9.5, 12.5, 16, 16, 18, 43.5}}]
```

95% confidence interval for 1 group:

Lower limit, Median, Upper limit, Initial number, $\alpha_{adj.} - \alpha_{calc.}$

Out[ ]= {{4.5, 9.5, 16, 11, 0.0114258}}

Supplementary matherial of the original: Mineral fertilization in tropical reforestation: seedling growth and survival of ten native tree species in Tabasco, Mexico.
Received: 24 September 2020 Accepted: 15 November 2021 Published: 17 June 2022

This document must be cited as:
Martínez-Bravo, R. D., Ricker, M., Nava-Cruz, Y., & Siebe, C. (2022). Mineral fertilization in tropical reforestation: seedling growth and survival of ten native tree species in Tabasco, Mexico. *Madera y Bosques*, *28*(1), e2812269. https://doi.org/10.21829/myb.2022.2812269. Supplement 2